%% @author Justin Sheehy <justin@basho.com>
%% @author Andy Gross <andy@basho.com>
%% @copyright 2007-2008 Basho Technologies
%%
%%    Licensed under the Apache License, Version 2.0 (the "License");
%%    you may not use this file except in compliance with the License.
%%    You may obtain a copy of the License at
%%
%%        http://www.apache.org/licenses/LICENSE-2.0
%%
%%    Unless required by applicable law or agreed to in writing, software
%%    distributed under the License is distributed on an "AS IS" BASIS,
%%    WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
%%    See the License for the specific language governing permissions and
%%    limitations under the License.

%% @doc Supervisor for the webmachine application.

-module(webmachine_sup).

-behaviour(supervisor).

%% External exports
-export([start_link/0, upgrade/0]).

%% supervisor callbacks
-export([init/1]).

-include("webmachine_logger.hrl").

%% @spec start_link() -> ServerRet
%% @doc API for starting the supervisor.
%% webmachine应用监督进程的启动
start_link() ->
	supervisor:start_link({local, ?MODULE}, ?MODULE, []).

%% @spec upgrade() -> ok
%% @doc Add processes if necessary.
upgrade() ->
	{ok, {_, Specs}} = init([]),
	
	Old = sets:from_list(
			[Name || {Name, _, _, _} <- supervisor:which_children(?MODULE)]),
	New = sets:from_list([Name || {Name, _, _, _, _, _} <- Specs]),
	Kill = sets:subtract(Old, New),
	
	sets:fold(fun (Id, ok) ->
					   supervisor:terminate_child(?MODULE, Id),
					   supervisor:delete_child(?MODULE, Id),
					   ok
			  end, ok, Kill),
	
	[supervisor:start_child(?MODULE, Spec) || Spec <- Specs],
	ok.

%% @spec init([]) -> SupervisorTree
%% @doc supervisor callback.
init([]) ->
	%% webmachine_router进程的启动配置
	Router = {webmachine_router,
			  {webmachine_router, start_link, []},
			  permanent, 5000, worker, [webmachine_router]},
	%% 启动日志进程
	LogHandler = [{webmachine_logger, {gen_event, start_link, [{local, ?EVENT_LOGGER}]},
				   permanent, 5000, worker, [dynamic]},
				  {webmachine_logger_watcher_sup, {webmachine_logger_watcher_sup, start_link, []},
				   permanent, 5000, supervisor, [webmachine_logger_watcher_sup]}],
	{ok, {{one_for_one, 9, 10},  LogHandler ++ [Router]}}.
